

```
function ustrich=f1(t,u)
ustrich=cos(t);
endfunction

function ustrich=f2(t,u)
ustrich=u;
endfunction

function ustrich=f3(t,u)
ustrich=-t*u;
endfunction

function u=explicitEuler(u0, t0, t, f)
u=zeros(size(t));

u(1)=u0;

for n=1:length(t)-1
    dt=t(n+1) - t(n);
    u(n+1)=u(n) + dt*f(t(n), u(n));
end
endfunction

function u=verbEuler(u0, t0, t, f)
u=zeros(size(t));

u(1)=u0;

for n=1:length(t)-1
    dt=t(n+1)-t(n);
    um=u(n) + dt/2*f(t(n), u(n));
    u(n+1)=u(n) + dt*f(t(n)+dt/2, um);
end
endfunction

// f1
T=12;
dt=0.5;
u0=0;

t=0:dt:T;

scilab_u=ode(u0, 0, t,f1);
explicitEuler_u=explicitEuler(u0, 0, t, f1);
verbEuler_u=verbEuler(u0, 0, t, f1);

scf(1), clf;
plot(t,scilab_u, 'b-*');
plot(t, u0+sin(t), 'k-');
plot(t,explicitEuler_u, 'r-*');
plot(t,verbEuler_u, 'g-*');

// f2
T=1.0;
dt=0.2;
u0=1;

t=0:dt:T;

scilab_u=ode(u0, 0, t,f2);
```

```
explicitEuler_u=explicitEuler(u0, 0, t, f2);
verbEuler_u=verbEuler(u0, 0, t, f2);

scf(2), clf
plot(t,scilab_u, 'b-*');
plot(t, u0*exp(t), 'k-');
plot(t,explicitEuler_u, 'r-*');
plot(t,verbEuler_u, 'g-*');

// f3
T=7.0;
dt=0.5;
u0=1;

t=0:dt:T;

scilab_u=ode(u0, 0, t,f3);
explicitEuler_u=explicitEuler(u0, 0, t, f3);
verbEuler_u=verbEuler(u0, 0, t, f3);

scf(3), clf
plot(t,scilab_u, 'b-*');
plot(t, u0*exp(-t.^2/2), 'k-');
plot(t,explicitEuler_u, 'r-*');
plot(t,verbEuler_u, 'g-*');
```





